

# NAG Fortran Library Routine Document

## F08FEF (SSYTRD/DSYTRD)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08FEF (SSYTRD/DSYTRD) reduces a real symmetric matrix to tridiagonal form.

### 2 Specification

```

SUBROUTINE F08FEF(UPLO, N, A, LDA, D, E, TAU, WORK, LWORK, INFO)
ENTRY      ssytrd (UPLO, N, A, LDA, D, E, TAU, WORK, LWORK, INFO)
INTEGER    N, LDA, LWORK, INFO
real     A(LDA,*), D(*), E(*), TAU(*), WORK(*)
CHARACTER*1 UPLO

```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine reduces a real symmetric matrix  $A$  to symmetric tridiagonal form  $T$  by an orthogonal similarity transformation:  $A = QTQ^T$ .

The matrix  $Q$  is not formed explicitly but is represented as a product of  $n - 1$  elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with  $Q$  in this representation (see Section 8).

### 4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: UPLO – CHARACTER\*1 *Input*

*On entry:* indicates whether the upper or lower triangular part of  $A$  is stored as follows:

if UPLO = 'U', the upper triangular part of  $A$  is stored;

if UPLO = 'L', the lower triangular part of  $A$  is stored.

*Constraint:* UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

3: A(LDA,\*) – **real** array *Input/Output*

**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .

*On entry:* the  $n$  by  $n$  symmetric matrix  $A$ . If UPLO = 'U', the upper triangular part of  $A$  must be stored and the elements of the array below the diagonal are not referenced; if UPLO = 'L', the lower triangular part of  $A$  must be stored and the elements of the array above the diagonal are not referenced.

- On exit:* A is overwritten by the tridiagonal matrix  $T$  and details of the orthogonal matrix  $Q$  as specified by UPLO.
- 4: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F08FEF (SSYTRD/DSYTRD) is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 5: D(\*) – *real* array *Output*  
**Note:** the dimension of the array D must be at least  $\max(1, N)$ .  
*On exit:* the diagonal elements of the tridiagonal matrix  $T$ .
- 6: E(\*) – *real* array *Output*  
**Note:** the dimension of the array E must be at least  $\max(1, N - 1)$ .  
*On exit:* the off-diagonal elements of the tridiagonal matrix  $T$ .
- 7: TAU(\*) – *real* array *Output*  
**Note:** the dimension of the array TAU must be at least  $\max(1, N - 1)$ .  
*On exit:* further details of the orthogonal matrix  $Q$ .
- 8: WORK(\*) – *real* array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1, LWORK)$ .  
*On exit:* if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.
- 9: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08FEF (SSYTRD/DSYTRD) is called, unless LWORK = -1, in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).  
*Suggested value:* for optimum performance LWORK should be at least  $N \times nb$ , where  $nb$  is the **blocksize**.  
*Constraint:*  $LWORK \geq 1$  or  $LWORK = -1$ .
- 10: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If  $INFO = -i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed tridiagonal matrix  $T$  is exactly similar to a nearby matrix  $A + E$ , where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$  is a modestly increasing function of  $n$ , and  $\epsilon$  is the **machine precision**.

The elements of  $T$  themselves may be sensitive to small perturbations in  $A$  or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

## 8 Further Comments

The total number of floating-point operations is approximately  $\frac{4}{3}n^3$ .

To form the orthogonal matrix  $Q$  this routine may be followed by a call to F08FFF (SORGTR/DORGTR):

```
CALL SORGTR (UPLO,N,A,LDA,TAU,WORK,LWORK,INFO)
```

To apply  $Q$  to an  $n$  by  $p$  real matrix  $C$  this routine may be followed by a call to F08FGF (SORMTR/DORMTR). For example,

```
CALL SORMTR ('Left',UPLO,'No Transpose',N,P,A,LDA,TAU,C,LDC,
+          WORK,LWORK,INFO)
```

forms the matrix product  $QC$ .

The complex analogue of this routine is F08FSF (CHETRD/ZHETRD).

## 9 Example

To reduce the matrix  $A$  to tridiagonal form, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix}.$$

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08FEF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, LDA, LWORK
PARAMETER       (NMAX=8,LDA=NMAX,LWORK=64*NMAX)
*      .. Local Scalars ..
INTEGER          I, INFO, J, N
CHARACTER        UPLO
*      .. Local Arrays ..
real            A(LDA,NMAX), D(NMAX), E(NMAX-1), TAU(NMAX-1),
+              WORK(LWORK)
*      .. External Subroutines ..
EXTERNAL         ssytrd
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08FEF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
READ (NIN,*) UPLO
IF (UPLO.EQ.'U') THEN
  READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
ELSE IF (UPLO.EQ.'L') THEN
  READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
END IF
*
*      Reduce A to tridiagonal form
```

```

*
      CALL ssytrd(UPLO,N,A,LDA,D,E,TAU,WORK,LWORK,INFO)
*
*      Print tridiagonal form
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Diagonal'
      WRITE (NOUT,99999) (D(I),I=1,N)
      WRITE (NOUT,*) 'Off-diagonal'
      WRITE (NOUT,99999) (E(I),I=1,N-1)
      END IF
      STOP
*
99999 FORMAT (1X,8F9.4)
      END

```

## 9.2 Program Data

```

F08FEF Example Program Data
  4                               :Value of N
  'L'                             :Value of UPLO
  2.07
  3.87 -0.21
  4.20  1.87  1.15
  -1.15  0.63  2.06 -1.81      :End of matrix A

```

## 9.3 Program Results

F08FEF Example Program Results

```

Diagonal
  2.0700  1.4741 -0.6492 -1.6949
Off-diagonal
  -5.8258  2.6240  0.9163

```

---